


Multivariate Difference Score Models for Body Dissatisfaction

Original Source: Gervais, S. J., Vescio, T. K., & Allen, J. (2011). When what you see is what you get: The consequences of the objectifying gaze for women and men. *Psychology of Women Quarterly*, 35(1), 5-17.

One minor question in this paper focused on gender differences in how discrepant the people felt their bodies were from their own ideals. To measure this, participants completed the Figure Rating Scale, which has pictures of nine bodies varying from extremely thin (0) to extremely overweight (10). Participants provided one rating for their *actual* body and another for their *ideal* body. Thus, the focus of the analysis was the difference score between these actual and ideal ratings for each person. Using a difference score as a DV is highly problematic, in that all information about the absolute amount of the original variables is lost. Here is a better way through a multivariate regression. These results use ML and differ slightly from actual results presented in the paper.

SAS Syntax and Output for Data Manipulation:

```
TITLE "Descriptive Info for Body Ratings by Gender";
PROC UNIVARIATE DATA=object_multiv; BY gender; VAR body5 body1;
    HISTOGRAM body5 body1 / NORMAL(COLOR=(blue) MU=est SIGMA=est)
        MIDPOINTS= -.5 to 10.5 by 1;
RUN;
```



Notice how the means and standard deviations differ by gender and rating type...

```
* Stack data into multivariate format (one row per person per DV);
DATA object_stacked; SET object_multiv;
  dv="Actual"; body=body5; OUTPUT;
  dv="Ideal "; body=body1; OUTPUT;
  LABEL body= "Body Size Rating (0=most thin)"; RUN;
```

Empty Multivariate Model Predicting Body Ratings (using Multivariate Normal Distribution and Identity Link): This model predicts the rating d for person $p \rightarrow Rating_{pd} = \beta_0 + \beta_1 RatingType_{pd} + e_{pd}$

```
TITLE "Empty Model: Differences between Actual and Ideal Figure Rating Scale";
PROC MIXED DATA=object_stacked COVTEST IC NAMELEN=50 METHOD=ML;
  CLASS dv;
  MODEL body = dv / SOLUTION DDFM=KR;
  REPEATED dv / R RCORR TYPE=UN SUBJECT=pin;
  ESTIMATE "Sample mean for Actual Rating" intercept 1 dv 1 0;
  ESTIMATE "Sample mean for Ideal Rating" intercept 1 dv 0 1; RUN;
```

Class Level Information			Iteration History			
Class	Levels	Values	Iteration	Evaluations	-2 Log Like	Criterion
dv	2	Actual Ideal	0	1	1140.87469005	
			1	1	819.23709010	0.00000000

Estimated R Matrix for Subject 1			Estimated R Correlation Matrix for Subject 1		
Row	Col1	Col2	Row	Col1	Col2
1	4.2638	1.8430	1	1.0000	0.8989
2	1.8430	0.9860	2	0.8989	1.0000

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	pin	4.2638	0.4923	8.66	<.0001
UN(2,1)	pin	1.8430	0.2251	8.19	<.0001
UN(2,2)	pin	0.9860	0.1138	8.66	<.0001

Fit Statistics	
-2 Log Likelihood	819.2
AIC (smaller is better)	829.2
AICC (smaller is better)	829.4
BIC (smaller is better)	844.3

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
819.2	5	829.2	829.4	835.4	844.3	849.3

Type 3 Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
dv	1	150	12.43	0.0006

Solution for Fixed Effects						
Effect	dv	Estimate	Standard Error	DF	t Value	Pr > t
Intercept		4.0267	0.08107	150	49.67	<.0001
dv	Actual	0.3600	0.1021	150	3.53	0.0006
dv	Ideal	0

Estimates					
Label	Estimate	Standard Error	DF	t Value	Pr > t
Sample mean for Actual Rating	4.3867	0.1686	150	26.02	<.0001
Sample mean for Actual Rating	4.0267	0.08107	150	49.67	<.0001

Multivariate Model Predicting Body Ratings (using Multivariate Normal Distribution and Identity Link)
 This model predicts the rating d for person p as a function of gender.

$$Rating_{pd} = \beta_0 + \beta_1 RatingType_{pd} + \beta_2 Gender_p + \beta_3 RatingType_{pd} * Gender_p + e_{pd}$$

```

TITLE1 "Differences between Actual and Ideal Figure Rating Scale";
TITLE2 "Sex Differences in Mean Ratings Only";
PROC MIXED DATA=object_stacked COVTEST IC NAMELEN=50 METHOD=ML;
  CLASS gender dv;
  MODEL body = gender|dv /SOLUTION DDFM=KR;
  REPEATED dv / R RCORR TYPE=UN SUBJECT=pin;
  LSMEANS gender dv gender*dv;
  ESTIMATE "How much bigger do women say they actually are than men?"
    gender -1 1 gender*dv -1 0 1 0 ;
  ESTIMATE "How much bigger do women say they ideally are than men?"
    gender -1 1 gender*dv 0 -1 0 1 ;
  ESTIMATE "What is the difference of actual and ideal size for men?"
    dv -1 1 gender*dv -1 1 0 0 ;
  ESTIMATE "What is the difference of actual and ideal size for women?"
    dv -1 1 gender*dv 0 0 -1 1 ;
RUN;
    
```

Class Level Information		
Class	Levels	Values
gender	2	Men Women
dv	2	Actual Ideal

Iteration History			
Iteration	Evaluations	-2 Log Like	Criterion
0	1	1136.04433438	
1	1	747.77758745	0.00000000

Convergence criteria met.

Estimated R Matrix for Subject 1		
Row	Col1	Col2
1	4.2050	1.8814
2	1.8814	0.9610

Estimated R Correlation Matrix for Subject 1		
Row	Col1	Col2
1	1.0000	0.9359
2	0.9359	1.0000

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	pin	4.2050	0.4855	8.66	<.0001
UN(2,1)	pin	1.8814	0.2248	8.37	<.0001
UN(2,2)	pin	0.9610	0.1110	8.66	<.0001

Fit Statistics	
-2 Log Likelihood	747.8
AIC (smaller is better)	761.8
AICC (smaller is better)	762.2
BIC (smaller is better)	782.9

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
747.8	7	761.8	762.2	770.3	782.9	789.9

Do the main effect and interaction with gender significantly improve the model?

Solution for Fixed Effects							
Effect	gender	dv	Estimate	Standard Error	DF	t Value	Pr > t
Intercept			3.8507	0.1198	150	32.15	<.0001
gender	Men		0.3179	0.1610	150	1.97	0.0501
gender	Women		0
dv		Actual	0.8060	0.1447	150	5.57	<.0001
dv		Ideal	0
gender*dv	Men	Actual	-0.8060	0.1945	150	-4.14	<.0001
gender*dv	Men	Ideal	0
gender*dv	Women	Actual	0
gender*dv	Women	Ideal	0

What is the reference DV and gender as determined by the CLASS statement?


Type 3 Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
gender	1	150	0.12	0.7293
dv	1	150	17.16	<.0001
gender*dv	1	150	17.16	<.0001

	Men	Women	Non-Gender
Actual	4.17	4.66	4.41
Ideal	4.17	3.85	4.01
??	4.17	4.25	

Estimates					
Label	Estimate	Standard Error	DF	t Value	Pr > t
How much bigger do women say they actually are than men?	0.4880	0.3368	150	1.45	0.1494
How much bigger do women say they ideally are than men?	-0.3179	0.1610	150	-1.97	0.0501
What is the difference of actual and ideal size for men?	-833E-17	0.1300	150	-0.00	1.0000
What is the difference of actual and ideal size for women?	-0.8060	0.1447	150	-5.57	<.0001

Least Squares Means							
Effect	gender	dv	Estimate	Standard Error	DF	t Value	Pr > t
gender	Men		4.1687	0.1640	150	25.42	<.0001
gender	Women		4.2537	0.1825	150	23.30	<.0001
dv		Actual	4.4127	0.1684	150	26.21	<.0001
dv		Ideal	4.0097	0.08050	150	49.81	<.0001
gender*dv	Men	Actual	4.1687	0.2251	150	18.52	<.0001
gender*dv	Men	Ideal	4.1687	0.1076	150	38.74	<.0001
gender*dv	Women	Actual	4.6567	0.2505	150	18.59	<.0001
gender*dv	Women	Ideal	3.8507	0.1198	150	32.15	<.0001

In words, how would you describe this interaction?



Multivariate Model Predicting Body Ratings (using Multivariate Normal Distribution and Identity Link)

This model predicts the rating d for person p as a function of gender, and allows different covariance matrices for each gender as well.

$$Rating_{pd} = \beta_0 + \beta_1 RatingType_{pd} + \beta_2 Gender_p + \beta_3 RatingType_{pd} * Gender_p + e_{pd}$$

```
TITLE1 "Differences between Actual and Ideal Figure Rating Scale";
TITLE2 "Sex Differences in Mean, Variance, and Covariance of Ratings";
PROC MIXED DATA=object_stacked COVTEST IC NAMELEN=50 METHOD=ML;
  CLASS gender dv;
  MODEL body = gender|dv / RESIDUAL SOLUTION DDFM=KR;

  * GROUP allows separate R matrices per gender, R= prints each group;
  REPEATED dv / R=1,84 RCORR=1,84 TYPE=UN SUBJECT=pin GROUP=gender;

  LSMEANS gender dv gender*dv;
  ESTIMATE "How much bigger do women say they actually are than men?"
    gender -1 1 gender*dv -1 0 1 0;
  ESTIMATE "How much bigger do women say they ideally are than men?"
    gender -1 1 gender*dv 0 -1 0 1;
  ESTIMATE "What is the difference of actual and ideal size for men?"
    dv -1 1 gender*dv -1 1 0 0;
  ESTIMATE "What is the difference of actual and ideal size for women?"
    dv -1 1 gender*dv 0 0 -1 1;

RUN;
```

Class Level Information		
Class	Levels	Values
gender	2	Men Women
dv	2	Actual Ideal

Iteration History			
Iteration	Evaluations	-2 Log Like	Criterion
0	1	1136.04433438	
1	1	741.26633941	0.00000000

Estimated R Matrix for Subject 1		
Row	Col1	Col2
1	4.8631	2.1884
2	2.1884	1.1282

Estimated R Correlation Matrix for Subject 1		
Row	Col1	Col2
1	1.0000	0.9343
2	0.9343	1.0000

Estimated R Matrix for Subject 84		
Row	Col1	Col2
1	3.3896	1.5010
2	1.5010	0.7538

Estimated R Correlation Matrix for Subject 84		
Row	Col1	Col2
1	1.0000	0.9390
2	0.9390	1.0000

Covariance Parameter Estimates						
Cov Parm	Subject	Group	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	pin	gender Men	4.8631	0.7549	6.44	<.0001
UN(2,1)	pin	gender Men	2.1884	0.3519	6.22	<.0001
UN(2,2)	pin	gender Men	1.1282	0.1751	6.44	<.0001
UN(1,1)	pin	gender Women	3.3896	0.5856	5.79	<.0001
UN(2,1)	pin	gender Women	1.5010	0.2679	5.60	<.0001
UN(2,2)	pin	gender Women	0.7538	0.1302	5.79	<.0001

Fit Statistics	
-2 Log Likelihood	741.3
AIC (smaller is better)	761.3
AICC (smaller is better)	762.0
BIC (smaller is better)	791.4

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
741.3	10	761.3	762.0	773.5	791.4	801.4

Does allowing different residual variances and covariance by gender significantly improve the model?

Solution for Fixed Effects							
Effect	gender	dv	Estimate	Standard Error	DF	t Value	Pr > t
Intercept			3.8507	0.1061	67	36.30	<.0001
gender	Men		0.3179	0.1576	150	2.02	0.0455
gender	Women		0
dv		Actual	0.8060	0.1305	67	6.17	<.0001
dv		Ideal	0
gender*dv	Men	Actual	-0.8060	0.1910	150	-4.22	<.0001
gender*dv	Men	Ideal	0
gender*dv	Women	Actual	0
gender*dv	Women	Ideal	0

Type 3 Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
gender	1	150	0.12	0.7242
dv	1	150	17.80	<.0001
gender*dv	1	150	17.80	<.0001

Estimates					
Label	Estimate	Standard Error	DF	t Value	Pr > t
How much bigger do women say they actually are than men?	0.4880	0.3304	150	1.48	0.1418
How much bigger do women say they ideally are than men?	-0.3179	0.1576	150	-2.02	0.0455
What is the difference of actual and ideal size for men?	9.1E-15	0.1395	83	0.00	1.0000
What is the difference of actual and ideal size for women?	-0.8060	0.1305	67	-6.17	<.0001

Least Squares Means							
Effect	gender	dv	Estimate	Standard Error	DF	t Value	Pr > t
gender	Men		4.1687	0.1767	83	23.59	<.0001
gender	Women		4.2537	0.1633	67	26.05	<.0001
dv		Actual	4.4127	0.1652	150	26.71	<.0001
dv		Ideal	4.0097	0.07881	150	50.88	<.0001
gender*dv	Men	Actual	4.1687	0.2421	83	17.22	<.0001
gender*dv	Men	Ideal	4.1687	0.1166	83	35.76	<.0001
gender*dv	Women	Actual	4.6567	0.2249	67	20.70	<.0001
gender*dv	Women	Ideal	3.8507	0.1061	67	36.30	<.0001

